Maple-assisted proof of formula for A221569

Robert Israel

4 June 2018

There are $5^2 = 25$ configurations for a sub-array of length 2. Consider the 25×25 transition matrix T such that $T_{ij} = 1$ if the left two elements of a sub-array of length 3 could be in configuration i while the right two elements are in configuration j (i.e. the middle entry is compatible with both i and j, and differs from at least one neighbour by something other than 1, and 0 otherwise. The following Maple code computes it.

```
> Configs:=  [seq(seq([i,j],i=0..4),j=0..4)]:
> Compatible:= proc(i,j)
    if Configs[i][2] <> Configs[j][1] then return 0 fi;
    if abs(Configs[i][2] - Configs[i][1]) = 1 and abs(Configs[j][1]
  -Configs[j][2]) = 1 then return 0 fi;
    1
  end proc:
> T:= Matrix(25,25,Compatible):
```

Thus for $n \geq 2$, $a(n) = u^T T^{n-1} u$ where u is a column vector with $u_i = 1$ for i corresponding to pairs that do not differ by 1, 0 otherwise. The following Maple code produces this vector.

```
> u:= Vector(25, i -> `if`(abs(Configs[i][1]-Configs[i][2]) = 1, 0,
  1)):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= u:
  for n from 1 to 15 do TV[n]:= T . TV[n-1] od:
> A:= [seq(u^%T . TV[n],n=0..15)];
```

$$A := [17, 59, 289, 1293, 5913, 26911, 122621, 558547, 2544357, 11590169, 52796369, 240501763, 1095550873, 4990531051, 22733220441, 103555975477] \quad \textbf{(1)}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{11} - 5\,t^{10} + 4\,t^9 - 6\,t^8 - 11\,t^7 - 9\,t^6 - 12\,t^5 - 4\,t^4 - 15\,t^3 - 3\,t^2 \quad \textbf{(2)}$$

This turns out to have degree 11, but with the two lowest coefficients 0. Thus we will have

$0 = u^T P(T)\, T^n\, u = \sum_{i=2}^{11} p_i\, a(i+n+1) = \sum_{i=0}^{9} p_{i+2}\, a(i+n+3)$ where p_i is the coefficient of t^i in $P(t)$.

That corresponds to a homogeneous linear recurrence of order 9, which would hold true for any u, after a delay of 3. It seems that with our particular u and v we have a recurrence of order only 5, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^2\,(t^2+t+1)\,(t^2-t+1)\,(t^5-5\,t^4+3\,t^3-t^2-15\,t-3) \quad \textbf{(3)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 6.

```
> Q:= t -> t^5 - 5*t^4 + 3*t^3 - t^2 - 15*t - 3:
  R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto (t^4+t^2+1)\,t^2 \quad \textbf{(4)}$$

Now we want to show that $b(n) = u^T\ Q(T)\ T^{n-1}\ u = 0$ for all $n \geq 1$. This will certainly satisfy the order-6 recurrence

$$\sum_{i=2}^{6} r_i\, b(i+n) = \sum_{i=2}^{6} r_i u^T\ Q(T)\ T^{n+i-1} v = u\ Q(T)\ R(T)\ T^{n-1}\ u = u\ P(T)\ T^{n-1}\ u = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(1) =\ldots = b(6) = 0$.

```
> seq(u^%T . Q(T) . T^(n-1) . u, n=1..6);
```

0, 0, 0, 0, 0, 0 **(5)**